

SAS Technical News

Autumn 2004

*For Higher
Customer Satisfaction,
We Bridge
the SAS System
Between
Customer's World.*

CONTENTS

- 1 SAS® 9の新機能と移行について

- 8 Q & A

- 14 新刊マニュアルのお知らせ

- 15 SASトレーニングのお知らせ

- 16 SASプログラマーズ・ワークショップ開催のお知らせ

- 16 最新リリース情報

特集

SAS® 9の 新機能と移行について

はじめに

2004年初夏から、ご利用を希望する全てのお客様に、SAS® 9が提供可能になりました。今号の特集では、以前からSASをご利用の方々が最も注目されている、下記の項目についてご紹介します。

- SAS9の新機能と拡張点
- 既存資産のSAS9での利用
- SAS9への移行方法

1. SAS9の新機能と拡張点

前号のSAS Technical News(2004年夏号)では、SAS9の概要についてご紹介しました。SAS9で追加された多くの新機能・拡張点から、今回はSASの中核となるBase SAS®を中心に、「マルチスレッド対応プロシジャ」、「SAS関数」、「マクロ機能」、「出力に関する新機能と拡張点」、および「他社製品とのデータ交換に関する新機能と拡張点」をご紹介します。

1.1. マルチスレッド対応プロシジャ

SAS9以前では、SASプログラムは単一CPU上の1つのプログラムとして実行されていました。SAS9では、下記の各プロシジャがマルチスレッド対応となり、複数CPU環境での処理分散が可能になったため、パフォーマンスの向上が期待できます。

Base SASのプロシジャ	MEANS, REPORT, SORT, SQL, SUMMARY, TABULATE
SAS/STATのプロシジャ	GLM, LOESS, REG, ROBUSTREG

1.2. SAS関数およびSAS CALLルーチン

SAS9では、新しいSAS関数やSAS CALLルーチンが追加されています。ここでは主なSAS関数とSAS CALLルーチンの例をご紹介します。

Perl正規表現を利用する関数

PRXではじまるいくつかの関数が追加され、SASでPerl正規表現が利用できるようになりました。これらの関数により、より柔軟性の高い文字列の検索・抽出や置換が行なえるようになります。次の例では、データの中から「数字2桁-(ハイフン)数字4桁-(ハイフン)数字4桁」に合致するデータを抽出するプログラムを、従来の方法と新しい関数を使用した方法で示しています。

従来のプログラム例:

```
DATA telnum;
  INPUT telnum $12.;
                                /* 数字と-のみであることを確認 */
  IF VERIFY(TRIM(telnum),'0123456789-')=0;
CARDS;
33-3343-4331
43-3323-1323
433-323-1323
33-3343-433X
;
RUN;

DATA telnum2;
  SET telnum;
                                /* パターン抽出 */
  WHERE telnum LIKE '____-____-____';
RUN;
```

従来では、パターン抽出にはIFステートメントとSUBSTR関数の組み合わせを多用するか、上記のプログラムのように、あらかじめSASデータセットを作成し、WHEREステートメントを利用しなければなりません。また、LIKE演算子では、文字値と数値の判別ができないため、VERIFY関数を併用するなどの工夫が必要でした。SAS9では、PRXMATCH関数により、パターン抽出を簡素化することができます。

PRXMATCH関数を使用した例:

```
DATA prxmatch;
  INPUT telnum $12.;
                                /* パターン抽出 */
  IF PRXMATCH('/\d{2}-\d{4}-\d{4}/',telnum);
CARDS;
33-3343-4331
43-3323-1323
433-323-1323
33-3343-433X
;
RUN;
```

また、文字列の置き換えを行なう場合は、PRXCHANGE関数を利用します。次の例では、文字列内にある「bat」、「hat」、「cat」の各文字列を「SAS9」に置換しています。

PRXCHANGE関数を使用した例:

```
DATA prxchange;
  txt1='hat, cat, bat !';
                                /* Xat(xはb,cまたはh)をSAS9に置き換える */
  txt2=PRXCHANGE('s/[bch]at/SAS9/',-1,txt1);
RUN;
```

文字列を結合する関数

CATではじまるいくつかの関数が追加され、文字列の結合が容易になりました。次の例では、変数NAME1とNAME2を結合して、FULLNAMEを作成しています。

従来のプログラム例:

```
DATA name;
  INPUT name1 $ name2 $;
  fullname1=name1 || name2;
  fullname2=TRIM(name1) || ' ' || TRIM(name2);
CARDS;
Mark Kobe
Bob Johnson
;
RUN;
```

従来のプログラムでは、文字列の結合を行なうには、結合演算子「||」を利用し、文字列の後方ブランクを削除するためにTRIM関数を併用するような処理が必要でした。SAS9では、CAT関数、CATX関数により文字列の結合が簡素化できます。

CATおよびCATX関数を使用した例:

```
DATA cat;
  INPUT name1 $ name2 $;
                                /* 変数名name1,name2を結合 */
  fullname1=CAT(of name1-name2);
                                /* 変数名name1,name2の後方ブランクを削除し */
                                /* 変数値にスペースを挿入 */
  fullname2=CATX(' ',of name1-name2);
CARDS;
Mark Kobe
Bob Johnson
;
RUN;
```

データセットCATの内容:

name1	name2	fullname1	fullname2
Mark	Kobe	Mark Kobe	Mark Kobe
Bob	Johnson	Bob Johnson	Bob Johnson

CATではしる関数では、変数リストを利用することができます。

文字・文字列のカウント

COUNT関数が追加され、文字列に含まれる任意の文字または文字列の出現回数を容易に求めることができるようになりました。次の例では変数内の「-」(ハイフン)と「33」の出現回数をカウントしています。

従来のプログラム例:

```
DATA cnt82;
  INPUT telnum $12.;
  _len=LENGTH(telnum);
  cnt1= LENGTH (telnum) - LENGTH (COMPRESS(telnum,'-'));
  cnt2=0;
  IF _len > 0 THEN
    DO _i=1 TO _len-1;
      _tmp=SUBSTR(telnum,_i,2);
      IF INDEX(_tmp,'33') > 0 THEN cnt2+1;
    END;
  DROP _:;
CARDS;
33-3343-4331
43-3323-1323
;
RUN;
```

上記のプログラムでは、文字にはLENGTH関数とCOMPRESS関数を利用して長さの差を求め、また文字列にはINDEX関数とDOループを併用して出現回数をカウントしています。SAS9では、こうした処理をCOUNT関数のみで容易に行なえます。

COUNT関数を利用した例:

```
DATA cnt91;
  INPUT telnum $12.;
  cnt1=COUNT(telnum,'-');
  cnt2=COUNT(telnum,'33');
CARDS;
33-3343-4331
43-3323-1323
;
RUN;
```

マクロの存在確認

SAS9から、マクロ変数の存在を確認するSYMEXIST関数が追加されました。これまでは、マクロ変数の存在を確認するには、sashelp.vmacroビューを参照するDATAステップ、またはSQLプロシジャステップを別途実行する必要がありました。次の例では、マクロ変数mvar1の存在を確認し、変数strへマクロ変数の内容を出力しています。

SYMEXIST関数の利用例:

```
%LET mvar1=test;

DATA mcheck91;
  IF SYMEXIST('mvar1') THEN DO;
    str=SYMGET("mvar1");
  END;
RUN;
```

マクロ作成時のブランク考慮

CALL SYMPUTXルーチンで、マクロ変数作成時の前後のブランクを考慮するプログラムを簡素化できるようになりました。これまでは、文字型変数の値を比較する際、前後にブランク(空値)が含まれる値が存在する場合にはLEFT関数やTRIM関数などを利用して、事前のデータ加工を考慮する必要がありました。SAS9では、この前後のブランクを考慮せずに、新たに追加されたCALL SYMPUTXルーチンだけで容易に文字型変数の値を比較できます。次の例では、変数varの値から文字列の前後のブランクを削除した値をマクロ変数に格納し、次のステップで他の変数との値を比較しています。

CALL SYMPUTXルーチンの利用例:

```
DATA mac91;
  var=" Macro Variable ";
  CALL SYMPUTX('macv1',var);
RUN;

DATA _null_;
  var="Macro Variable A";
  IF var=%&macv1 A THEN PUT 'HIT 1';
RUN;
```

1.3. 出力(OUTPUT GRAPH)に関する新機能と拡張点

出力形式 / 入力形式

出力形式 / 入力形式名の長さが、以下のように拡張されました。

項目	SAS9での長さ	SAS9以前の長さ
数値フォーマット	32バイト	8バイト
文字フォーマット (必ず\$で始める必要あり)	32バイト	8バイト
数値インフォーマット	31バイト	7バイト
文字インフォーマット (必ず\$で始める必要あり)	31バイト	7バイト

VIEWTABLEでは、長さの制約は以前と同様です。

なお、新しく追加されたVALIDFMTNAME=FAILシステムオプションを使用すると、前バージョンの出力形式 / 入力形式名の長さを超えるものを作成・使用した際に、ログにエラーメッセージを出力することができます。

SAS/GRAPHに追加されたプロシジャ

SAS9では新たにGBARLINE、GAREAVAR、GMAPIMPORTの3つのプロシジャが追加されています。ここではGBARLINEプロシジャとGAREAVARプロシジャをご紹介します。

GBARLINEプロシジャ

これまで、複合グラフを作成するには、GCHARTプロシジャのANNOTATEオプションを利用するか、GPLOTプロシジャのOVERLAYオプションを利用する必要がありました。SAS9では、GBARLINEプロシジャで容易に複合グラフを作成できます(ただし、グラフ描画の際に使用できるデバイスに制限があります。詳細はオンラインヘルプを参照してください)。次の例では、月別の売上実績の棒グラフと売り上げ予測の折れ線グラフをGBARLINEプロシジャのみで描画しています。

GBARLINEプロシジャの使用例:

```

/* SAMPLEデータ作成 */
DATA gbarline;
  SET sashelp.prdsal3;
  WHERE country='カナダ' AND year=1998 AND product='机';
  mon=MONTH(date);
  LABEL mon='月';
RUN;
PROC SORT DATA=gbarline;
  BY mon;
RUN;

GOPTIONS RESET=ALL DEVICE=activex;
ODS HTML FILE='c:\%gbarline.htm';
TITLE1 "1998年売上レポート 国:カナダ 製品:机";
PROC GBARLINE DATA=gbarline;
  BAR mon / DISCRETE SUMVAR=actual;
  PLOT / SUMVAR=predict;
RUN;
QUIT;
ODS HTML CLOSE;
    
```




図1:GBARLINE出力例

GAREAVARプロシジャ

GAREAVARプロシジャを利用することで、評価される2つの変数の項目を、棒グラフの領域として視覚的に確認できるようになりました(グラフ描画の際に使用できるデバイスに制限があります。詳細は、オンラインヘルプを参照してください)。次の例では、冬季オリンピックにおけるメダル種別ごとの獲得数の割合を、国別に表示しています。

GAREAVARプロシジャの使用例:

```

GOPTIONS RESET=ALL DEVICE=activex;
ODS HTML FILE='C:\%gareavar.htm';
/* SAMPLEデータ作成 */

DATA medales;
  INPUT country $15. medaltyp $ winter summer athletes;
DATALINES;
Germany      Gold   12 14 176
Germany      Silver 16 17   0
Germany      Bronze  7 26   0
United States Gold  10 39 210
United States Silver 13 25   0
United States Bronze 11 33   0
Norway       Gold   11  4  42
Norway       Silver  7  3   0
Norway       Bronze  6  3   0
Canada       Gold    6  3 157
Canada       Silver  3  3   0
Canada       Bronze  4  8   0
Russia       Gold    6 32 160
Russia       Silver  7 28   0
Russia       Bronze  3 28   0
;

PROC GAREABAR DATA=medals;
  HBAR country*athletes /SUMVAR=winter
                        SUBGROUP=medaltyp
                        WSTAT=percent
                        RSTAT=percent;
RUN;
QUIT;

ODS HTML CLOSE;
    
```




図2:GAREAVARプロシジャ出力例

1.4. 他社製品とのデータ交換に関連する新機能と拡張点

Microsoft Excel IMPORT / EXPORT プロシジャ

IMPORT プロシジャとEXPORT プロシジャで、データセットオプションが利用可能になりました。サンプルプログラムを本号のQ&Aのページでご紹介していますので、そちらをご参照ください。

Microsoft Excel / Microsoft Accessへの

LIBNAMEステートメント / SQLパススルー

LIBNAMEステートメントのエンジンに、EXCELとACCESSが追加されました。また、SQLパススルーでのアクセスも可能となりました。詳細は、前号のSAS Technical News (2004年夏号)でご紹介していますので、そちらをご参照ください。

2. 既存資産の利用

SAS9ではSAS8同様、既存資産を利用することが可能です。ここでは、既存資産を利用する際の注意点・仕様についてご紹介します。

2.1. 前リリースがインストール済みの環境にSAS9をインストールする際の注意点

前リリースのSASがインストールされた環境にSAS9をインストールする際には、下記の点にご注意ください。

あくまでもSAS9への切り替えを行なう上での一過程としてお考えください。決して前リリースのSASとSAS9を同一環境にインストールすることを推奨している訳ではありません。

インストールディレクトリ

インストールディレクトリは、前リリースとSAS9を、完全に分けてください。

ファイル拡張子について

SAS8から、SASファイルの拡張子に変更されています。下記は、ファイルの種類と拡張子の対照表です。

ファイルの種類	SAS6での拡張子	SAS8, SAS9での拡張子
LOG	.LOG	.LOG
OUTPUT	.LST	.LST
PROGRAM	.SAS	.SAS
ACCESSディスクリプタ	.SA2	.SAS7BACS
SASカタログ	.SC2	.SAS7BCAT
SASデータセット	.SD2	.SAS7BDAT
テーブルインデックス	.SI2	.SAS7BNDX
データセットビュー	.SV2	.SAS7BVIEW
MDDDB	.SM2	.SAS7BMDB
ストアドプログラム	.SS2	.SAS7BPGM

SAS6での拡張子は、Windows版SASのもので、

SAS8以降では、すべてのプラットフォームにて同様の拡張子となります。

システム必要条件の変更

SAS9のシステム必要条件に適合しているコンピュータをご使用ください。Windows版SAS9は、「Windows 9x系のOS」をサポートしていません。また、UNIX版SAS9(Linux版を除く)は、64bit版のみとなっていますので、特にご注意ください。システム必要条件の詳細に関しては、下記URLをご参照ください。

「インストールセンター」

<http://www.sas.com/japan/service/installcenter/index.html>

Windows版SASの注意点

- Windows版SAS8がインストールされている環境にSAS9をインストールすると、SAS Shared Filesが上書きされます。そのため、ODSで使用するActive X、Javaアプレット、拡張エディタなどのモジュールが上書きされます。SAS8から利用するこれらのモジュールは、SAS9のものになります。
- SAS9をインストールすると、Microsoft Data Access Components (MDAC) がアップグレードされます。SAS/ACCESSのOLE DBインターフェイス、またはPC File Formatsを使用するSAS8を共用する場合は、SAS9をインストールする前にHot Fix(修正ファイル)を適用してください。なお、Hot Fixの詳細に関しては、弊社テクニカルサポートまでお問い合わせください。

2.2. 利用時の注意点

V6・V8形式ファイルの取り扱いについて

NativeとNon-Nativeによって、参照・編集の仕様が異なります。以下に該当する場合は、Non-Nativeです。

- OSファミリーが異なる
例:Windows版SAS6で作成したデータセットをAIX版SAS9で利用する
- 利用するSASのbitが異なる
例:Solaris版SAS6(32bit)で作成したデータセットをSolaris版SAS9(64bit)で利用する
- ファイルが作成された際のエンコードが異なる
例:Solaris版SAS6 SJISで作成したデータセットをSolaris版SAS9 EUCで利用する

V6形式ファイルのSAS9での扱い方[R:参照 W:書き込み U:更新]

ファイルタイプ	Native	Non-Native
SASデータセット	R / W / U	R 1
DATAステップビュー	R	Not Supported
SQLビュー	R	Not Supported
SAS/ACCESSビュー	R 2	Not Supported
SASカタログ	R	Not Supported

- 同一OSファミリーおよびエンコードで作成されたファイルのみがサポート対象です。
- SAS/ACCESSビューは、SAS8以降の新機能を利用することができません。

V8形式ファイルのSAS9での扱い方[R:参照 W:書き込み U:更新]

ファイルタイプ	Native	Non-Native
SASデータセット	R / W / U	R / W 3
DATAステップビュー	R / W / U	Not Supported
SQLビュー	R / W / U	R
SAS/ACCESSビュー	R / W / U 4	R 5
SASカタログ	R / W / U	Not Supported

- 同一エンコードで作成されたファイルのみがサポート対象です。また、SAS9は、32bit環境で作成されたデータセットを使って同一名のデータセットを作成すると、デフォルトでは64bitで作成します。
- SAS/ACCESSビューは、SAS8以降の新機能を利用することができません。
- SAS/ACCESS OracleインターフェイスおよびSybaseインターフェイスによって作成されたファイルのみがサポート対象です。

格納方法

SASファイルの管理方法はディレクトリ単位となります。バージョンごとに格納ディレクトリを分ける必要があります。

LIBNAME定義例:

```
LIBNAME libref V6 'SAS6のデータセット・カタログの格納ディレクトリ名' ;
LIBNAME libref V8 'SAS8のデータセット・カタログの格納ディレクトリ名' ;
LIBNAME libref V9 'SAS9のデータセット・カタログの格納ディレクトリ名' ;
```

エンジン指定しなかった場合、SAS9は以下のようにエンジンを自動的に割り当てます。

	割り当てエンジン
ディレクトリにファイルが一切存在しない場合	V9
ディレクトリにSAS6のファイルのみ存在する場合	V6
ディレクトリにSAS8のファイルのみ存在する場合	V9
ディレクトリにSAS9のファイルのみ存在する場合	V9
ディレクトリにSAS6・SAS9の両方のファイルが存在する場合	V9

SAS9以前のバージョンから、SAS9で作成したデータセットを利用するSAS6から、SAS8およびSAS9のデータセットを直接参照することはできません。Native、Non-Nativeにより、対応方法は以下のように異なります。

Nativeの場合

Nativeの場合は、SAS9でLIBNAMEステートメントにV6エンジンを指定してライブラリを割り当て、そのライブラリに生成したデータセットをSAS6から利用することが可能です。下記にプログラム例を示します。

```
LIBNAME sasdata V6 'd:¥sasdata' ;
DATA sasdata.v6data ;
  x = 'V6 data' ;
RUN ;
```

上記をSAS9で処理した場合(D:\SASDATAにはファイルが存在しない)、「SASDATA」ライブラリはV6エンジンになるので、SASデータセット「V6DATA」もV6エンジンで生成されます。つまりV6でも認識することができます。

Non-Nativeの場合

Non-Nativeの場合は、SAS9でLIBNAMEにXPORTエンジンを指定して移送ファイルを作成し、SAS6で移送ファイルを復元します。下記にプログラム例を示します。

SAS9で移送ファイルを作成:

```
OPTIONS VALIDVARNAME=V6 VALIDFMTNAME=FAIL;
LIBNAME tran XPORT '/data/tran.xpt'; /* 移送ファイル */
PROC COPY IN =v9lib OUT=tran MT=data;
RUN;
```

SAS6で移送ファイルを復元:

```
OPTIONS VALIDVARNAME=V6 VALIDFMTNAME=FAIL;
LIBNAME tran XPORT '/data/tran.xpt'; /* 移送ファイル */
PROC COPY IN =tran OUT=v6lib;
RUN;
```

なお、SAS8からSAS9のデータセットを参照することが可能です。ただし、SAS9の新機能を利用して作成したデータセットは、SAS8から参照できないのでご注意ください。

3. SAS9への移行について

「2. 既存資産の利用」に記載したように、SAS9から以前のリリースで作成された資産を活用することは可能です。ただし、SAS9の新機能を利用するためには資産を移行する必要があります。ここではその移行方法をご紹介します。

3.1. MIGRATEプロシジャ

SAS6からSAS8に移行する際は、COPYプロシジャまたはCPORTプロシジャ / CIMPORTプロシジャを利用しました。SAS9でもそれらを利用することは可能ですが、新たに追加されたMIGRATEプロシジャの利用で、移行作業が容易になりました。

MIGRATEプロシジャの入力SAS形式ファイルはSAS 6.12以降で作成されている必要があります。

MIGRATEプロシジャの概要

MIGRATEプロシジャを使用してSAS9にファイルを移行するには、下記の条件を満たしている必要があります。

- ・移行するSAS形式ファイルが、SAS 6.12以降で作成されている
- ・同一OSファミリー間での移行である

また、下記のファイル類は、MIGRATEプロシジャでは移行できません。

- ・AIX版SAS6で作成されたSASCatalog
- ・NOSAVEオプションで作成されたデータステップビュー (ただし、SAS6は不可)
- ・SAS/ACCESS OracleインターフェイスおよびSybaseインターフェイス以外で作成したSAS/ACCESS ビュー
- ・マクロカタログ
- ・スタッドプログラム

MIGRATEプロシジャの使用条件に合致しない場合は、従来のCOPYプロシジャまたはCPORTプロシジャ / CIMPORTプロシジャを利用する必要があります。「3.2.CPORT / CIMPORTプロシジャ」を参照してください。

MIGRATEプロシジャの構文は、以下の通りです。

```
LIBNAME source エンジン名 "パス";
LIBNAME target V9 "パス";
PROC MIGRATE IN=source OUT=target <OPTION>;
RUN ;
```

Nativeファイルの移行

MIGRATEプロシジャでのNativeファイルの移行は、以下の手順で行ないます(Nativeファイルの詳細は、「2.2. 利用時の注意点」をご参照ください)。

1) 空のディレクトリを用意し、V9エンジンのSASライブラリを定義する

```
LIBNAME v9lib V9 "d:¥saslib¥v9lib";
```


- 2) MIGRATE プロシジャを使用して、V9エンジンのSASライブラリにV6またはV8エンジンのデータセット等を移行する

```
LIBNAME v8lib V8 "d:\saslib\v8lib"; /* V8エンジンのSASライブラリ */
PROC MIGRATE IN=v8lib OUT=v9lib ;
RUN ;
```




図3: MIGRATE プロシジャによるNativeのファイル移行

Non-Nativeファイルの移行

MIGRATE プロシジャでのNon-Nativeファイルの移行には、リモートライブラリサービス(RLS)を利用します(Non-Nativeファイルの詳細は、「2.2 利用時の注意点」をご参照ください)。

- ・ RLSを利用するには、SAS/CONNECT[®]またはSAS/SHARE[®]のライセンスが必要です。これらのプロダクトを導入していない場合には、次項「CPORT/CIMPORT プロシジャ」をご参照ください。
- ・ 以下の例は、同一筐体にSAS8とSAS9がインストールされていることを前提にしています。

RLSの利用は、大量データには不向きです。

- 1) SAS8のリモートホストにて、移行入力ライブラリを定義する

```
LIBNAME v8srv V8 "/DATA/v8lib";
```

- 2) SAS9で、手順1で定義した同一ディレクトリに異なるライブラリ参照名を定義する

```
LIBNAME source V8 "/DATA/v8lib";
```

- 3) リモートライブラリサービス(RLS)を定義する

```
LIBNAME v8srv SERVER=unix1;
```

- 4) 空のディレクトリを用意し、V9エンジンのSASライブラリを定義する

```
LIBNAME v9lib V9 "/DATA/v9lib";
```

- 5) MIGRATE プロシジャを使用して、V9エンジンのSASライブラリにV6またはV8エンジンのデータセット等を移行する

```
PROC MIGRATE IN=source OUT=v9lib SLIBREF=v8srv;
RUN ;
```




図4: MIGRATE プロシジャによるNon-Nativeのファイル移行

3.2. CPORT / CIMPORT プロシジャ

異種OSファミリー間など、MIGRATE プロシジャを利用できない場合は、従来と同様にCPORT プロシジャ、CIMPORT プロシジャを利用します。なお、CPORT プロシジャ、CIMPORT プロシジャは、データセットおよびカタログのみがサポート対象です。

- 1) SAS6またはSAS8で、CPORT プロシジャを使用して移送ファイルを作成する

```
FILENAME v8cat "/DATA/v8cat.cpt" ; /* 移送ファイル */
LIBNAME v8lib V8 "/DATA/v8cat"; /* エンジン指定 */
PROC CPORT LIB=v8lib FILE=v8cat MEMTYPE=CATALOG;
RUN ;
```

- 2) 空のディレクトリを用意し、V9エンジンのSASライブラリを定義し、CIMPORT プロシジャで移送ファイルを復元する

```
LIBNAME v9lib V9 "/DATA/v9lib" ; /* V9エンジンのSASライブラリ */
FILENAME v8cat "/DATA/v8cat.cpt"; /* 移送ファイル */
PROC CIMPORT LIB=v9lib INFILE=v8cat ;
RUN ;
```

4. おわりに

今回ご紹介しきれなかった新機能や、移行上の注意点について、「SAS プログラマーズ・ワークショップ」を開催してご紹介する予定です。「SAS プログラマーズ・ワークショップ」の開催概要については、16ページをご参照ください。なお、SAS9には、ここでご紹介した以外にも数多くの新機能・拡張点があります。また、各種の新製品もリリースされています。弊社および米国本社のWebサイトや関連マニュアル、またオンラインヘルプをご参照いただき、SAS9をご活用ください。

Q&A



- TABULATEプロシジャで列の合計に対するパーセンテージを求める方法
- IMPORTプロシジャで変数名を指定する方法
- SAS/ACCESSで外部データベースのテーブルにある1024バイト以上の文字型データを取り込む方法
- 移送形式ファイルでの移送で数値変数の長さが異なる場合がある
- ODSで作成された既存のHTMLファイルに追記する方法
- マクロプログラムにおけるネスト情報の出力
- SASシステムオプションの保存と復元について
- ログを外部ファイルに保存する際のファイル名指定について
- 改行コードがない可変長データを読み込む方法
- Hodrick-PrescottフィルターとBaxter-Kingフィルターについて
- 正確な並べ替え分布に基づくCochran-Armitageの傾向性検定

Q TABULATEプロシジャで下記のような表を作りました。PCTN(度数のパーセンテージ)は、全件を100%とした値になりますが、列を100%とした値にするには、どのように指定すればいいですか。

```

-----
|               |               |               |               |
|               |               |               |               |
|               |               |               |               |
|               |               |               |               |
|               |               |               |               |
|-----+-----+-----+-----|
|group      |gender      |               |               |
|-----+-----+-----+-----|
|A          |男          |               |               |
|          |-----+-----+-----+-----|
|          |女          |               |               |
|          |-----+-----+-----+-----|
|          |All         |               |               |
|-----+-----+-----+-----|
|B          |gender      |               |               |
|          |-----+-----+-----+-----|
|          |男          |               |               |
|          |-----+-----+-----+-----|
|          |女          |               |               |
|          |-----+-----+-----+-----|
|          |All         |               |               |
|-----+-----+-----+-----|
|All        |               |               |               |
|-----+-----+-----+-----|
    
```

上記の表を作成したプログラム:

```

PROC FORMAT;
  VALUE $genderf
    '1' = '男',
    '2' = '女';
;
  VALUE ynf
    1 = 'はい',
    2 = 'いいえ';
;
RUN;

DATA sample;
  INPUT group $ gender $ a1 @@;
  FORMAT gender genderf. a1 ynf.;
CARDS;
A 1 2   A 1 1
A 1 1   A 1 1
A 2 2   A 2 1
B 1 1   B 2 2
B 2 1   B 1 2
;
RUN;

PROC TABULATE DATA=sample;
  CLASS group gender a1;
  TABLES (group*(gender ALL)) ALL,a1*PCTN
  / MISSTEXT='0' PRINTMISS;
;
RUN;
    
```

A SAS8から追加された統計量キーワードCOLPCTNを使用して、列に対するパーセンテージを求められます。

```

PROC TABULATE DATA=sample;
  CLASS group gender a1;
  TABLES (group*(gender ALL)) ALL,a1*COLPCTN
  / MISSTEXT='0' PRINTMISS;
;
RUN;
    
```

SAS8では、度数や合計に関して、下記の統計量キーワードが追加されています。

COLPCTN	列
PAGEPCTN	ページ
REPPCTN	表
ROWPCTN	行
COLPCTSUM	列
PAGEPCTSUM	ページ
REPPCTSUM	表
ROWPCTSUM	行

なお、SAS6で列を100%とした表を出力するには、次のようなTABLESステートメントを記述します。

```
TABLES (group*(gender ALL)) ALL,a1*PCTN<group*gender group*ALL ALL>
```

列を100%とした表の出力:

		a1	
		はい	いいえ
		ColPctN	ColPctN
group	gender		
A	男	50.00	25.00
	女	16.67	25.00
	All	66.67	50.00
B	gender		
	男	16.67	25.00
	女	16.67	25.00
	All	33.33	50.00
All		100.00	100.00

Q 1行目に列見出しのないExcelデータをIMPORTプロシジャで読み込むと、変数名が自動的に「F1,F2,...」のようになってしまいます。読み込む際に変数名を指定する方法はありますか。

A SAS 8.2までは、変数名を変更するには、いったんIMPORTプロシジャで読み込んだ後、DATASETSプロシジャなどで新しい変数名を指定する必要がありました。SAS9からは、IMPORTプロシジャ内でデータセットオプションを使用できるようになりました。データセットオプションのRENAMEオプションを使用することで、読み込みと同時に変数名を定義することができます。下記の例を参照してください。

```
PROC IMPORT DATAFILE="c:\mydir\Book1.xls"
  OUT=sheet1(RENAME=(F1=Name F2=Age F3=Weight F4=Height));
  SHEET='Sheet1';
  GETNAMES=no;
RUN;
```

Q SAS/ACCESSを利用してデータベースのテーブルを読み込む際に、文字型データが1024バイトで切り捨てられてしまいます。切り捨てずにすべてを読み込むにはどうすればよいでしょうか。

A SAS/ACCESSのLIBNAMEエンジンおよびSQLパススルー機能を利用してデータベースのテーブルから文字型データを読み込む際の、文字長のデフォルト値は1024バイトです。この値を変更するには、DBMAX_TEXTオプションを利用します。指定可能な値の範囲は1から32767までです。

プログラム例:

```
PROC SQL ;
  CONNECT TO DB2
    (USER=***** USING=*****
     DB=***** SCHEMA=*****
     DBMAX_TEXT=32767);
  (中略)
QUIT;
```

なお、DBMAX_TEXTの設定を変更すると、文字型データはすべて変更後の長さで読み込まれます。

Q 移送形式ファイルを使用してUNIXからWindowsにデータセットを移送する場合、8バイト未満の数値変数が定義されていると自動的に1バイト追加されてしまいます。この現象を回避する方法はありますか？

A SASでは、移送形式ファイルにおいて8バイト未満の数値変数が定義されている場合、特に移送元と移送先のOSが異なる場合などに発生する数値精度の違いにより数値精度が失われることを防ぐために、数値変数の長さを1バイト増やします。これは、同一OS間での移送においても同様の現象となります。この動作仕様については、下記URLでご紹介しています。

「Technical FAQ(1043)」

<http://support.sas.com/faq/010/FAQ01043.html>

ただし、XPORT形式ではなく、CPORTプロシジャによるCPORT形式で移送ファイルを作成した場合には、CIMPORTプロシジャのオプションであるEXTENDSN=オプションを指定すれば、数値変数の長さが1バイト追加されることを回避できます。

```
PROC CIMPORT LIB=mylib FILE="C:\mydir\sample.cpt"
  EXTENDSN=NO;
RUN;
```

Q ODS HTMLステートメントでHTMLファイルを出力する場合、既存ファイルと同じ出力ファイル名を指定すると、内容が上書きされます。このとき、ファイルを上書きせず、追記する方法を教えてください。

A 次のオプションとサブオプションを、FILENAMEステートメントやODS HTMLステートメントで指定することで、ODS出力を単一のHTMLファイルに追記することが可能です。

MODオプション

FILENAMEステートメントのオプションです。出力を既存のファイルに追記します。

NO_BOTTOM_MATTER、NO_TOP_MATTERサブオプション

ODS HTMLステートメントにおいてFILE=オプションまたはBODY=オプションとともに使用するサブオプションです。NO_BOTTOM_MATTERを指定するとHTML定義終了のタグが省略され、NO_TOP_MATTERを指定するとHTML定義開始のタグが省略されます。それぞれNOBOT、NOTOPと略することができます。

以下のサンプルプログラムでは、3つのPROCステップの結果を単一のHTMLファイル(sample.htm)に出力しています。上記オプション、サブオプションの使用方法についてご確認ください。

```
FILENAME sample 'C:\temp\sample.htm' ;
ODS HTML FILE=sample (nobot) ;
PROC PRINT DATA=sashelp.class ;
RUN ;
ODS HTML CLOSE ;

FILENAME sample 'C:\temp\sample.htm' MOD ;
ODS HTML FILE=sample (notop nobot) ;
PROC FREQ DATA=sashelp.class ;
    TABLES sex ;
RUN ;
ODS HTML CLOSE ;

ODS HTML FILE=sample (notop) ;
PROC UNIVARIATE DATA=sashelp.class ;
RUN ;
ODS HTML CLOSE ;
```

Q マクロプログラム中で他のマクロプログラムを呼び出す際、MPRINTシステムオプションやMLOGICシステムオプションを設定しても、実行中のマクロプログラムがどのマクロプログラムから呼び出されているのか判断できません。何か良い方法はありますか。

A SAS9で追加されたMPRINTNESTおよびMLOGICNESTシステムオプションにより、実行されたマクロプログラムがどのマクロプログラムから呼び出されたのか(いわゆるネストの情報)が、MPRINTおよびMLOGICシステムオプションの情報として実行ログに表示されます。以下は、同じマクロプログラムをSAS8とSAS9でそれぞれ実行した際のログです。MPRINTおよびMLOGICシステムオプションが出力する情報の違いをご確認ください。

SAS8での実行ログ:

```
1  %MACRO a;
2  DATA _null_;
3      PUT 'hello A';
4  RUN;
5  %MEND;
6  %MACRO b;
7  %a
8  DATA _null_;
9      PUT 'hello B';
10 RUN;
11 %MEND;
12
13 OPTIONS MPRINT MLOGIC;
14
15 %b
MLOGIC(B): 実行を開始します。
MLOGIC(A): 実行を開始します。
MPRINT(A): DATA _null_;
MPRINT(A): PUT 'hello A';
MPRINT(A): RUN;
```

```
hello A
NOTE: DATA ステートメント 処理 :
      処理時間          0.10 秒
      CPU 時間          0.00 秒
```

```
MLOGIC(A): 実行を終了します。
MPRINT(B): DATA _null_;
MPRINT(B): PUT 'hello B';
MPRINT(B): RUN;
hello B
NOTE: DATA ステートメント 処理 :
      処理時間          0.00 秒
      CPU 時間          0.00 秒
```

```
MLOGIC(B): 実行を終了します。
```

SAS9での実行ログ:

```
1  %MACRO a;
2  DATA _null_;
3      PUT 'hello A';
4  RUN;
5  %MEND;
6
7  %MACRO b;
8  %a
9
10 DATA _null_;
11     PUT 'hello B';
12     RUN;
13 %MEND;
```

```

14
15  OPTIONS MPRINT MPRINTNEST MLOGIC MLOGICNEST;
16
17  %b
MLOGIC(B): 実行を開始します。
MLOGIC(B.A): 実行を開始します。
MPRINT(B.A):  DATA _null_;
MPRINT(B.A):  PUT 'hello A';
MPRINT(B.A):  RUN;

```

hello A

NOTE: DATA ステートメント 処理 (合計処理時間):

処理時間	0.02 秒
CPU 時間	0.03 秒

MLOGIC(B.A): 実行を終了します。

```

MPRINT(B):  DATA _null_;
MPRINT(B):  PUT 'hello B';
MPRINT(B):  RUN;

```

hello B

NOTE: DATA ステートメント 処理 (合計処理時間):

処理時間	0.00 秒
CPU 時間	0.00 秒

MLOGIC(B): 実行を終了します。

Q

SASシステムオプションの設定値を、必要に応じて変更して
います。設定のリセットを簡単に行なう方法はありますか。

A

SAS 8.2から、OPTSAVEプロシジャおよびOPTLOADプロ
シジャが追加され、オプションの設定値を保存、復元できる
ようになりました。これらのプロシジャの使用法は、下記のサ
ンプルプログラムを参考にしてください。

```

PROC OPTIONS OPTION=missing;          /* 現状確認 */
RUN;
PROC OPTIONS OPTION=ls;
RUN;

PROC OPTSAVE OUT=sasuser.defaultopts; /* 保存 */
RUN;

OPTIONS missing='M';                  /* 変更 */
OPTIONS ls=100;

PROC OPTIONS OPTION=missing;          /* 変更確認 */
RUN;
PROC OPTIONS OPTION=ls;
RUN;

PROC OPTLOAD DATA=sasuser.defaultopts; /* 復元 */
RUN;

```

```

PROC OPTIONS OPTION=missing;          /* 復元確認 */
RUN;
PROC OPTIONS OPTION=ls;
RUN;

```

Q

バッチモードでSASを利用する際、実行ログファイル名に日
付や時間などを含めることはできますか。

A

SAS 8.2の場合、実行ログのファイル名を任意に指定する
には、-LOGシステムオプションでファイル名を明示的に指
定する必要があります。つまり、日付などを付加した
「saslog_yyyymmdd.log」というファイル名で実行ログを保存するには、
オペレーティングシステムの機能などを利用して、あらかじめファイル名を
指定します。

たとえばWindowsでは、取得した日付をファイル名に含めるには、次のよ
うなコマンドバッチを記述する必要があります。

```

@ECHO OFF
SET
LOGNAME="saslog_%date:~2,4%%date:~7,2%%date:~10,2%.log"
"C:\Program Files\SAS Institute\SAS\%8%nl%ja\sas.exe"
"program.sas" -log %logname%

```

LOGNAME=の行は、改行せずに1行で記述する必要があります。

SAS 9.1では、実行ログファイル名の作成機能が追加され、このような作
業が容易になりました。たとえば、「saslog_yyyymmdd.log」というファイ
ル名は、-LOG、-LOGPARMシステムオプションを利用して、以下のように
記述することができます。

```
-log "saslog_#Y#m#d.log" -logparm "rollover=session"
```

この他にも、ログファイルに関する多数のオプションが用意されています。
詳細については、オンラインドキュメントなどをご参照ください。

Q レコード内に各レコード項目の識別子はあるが、レコードの長さが記載されたフィールドがない、つまり「改行コードのない」可変長のテキストファイルがあります。このようなファイルをSASで読み込むには、どのようにすればよいでしょうか。

A 改行コードが存在しない可変長データの代表的なケースには、メインフレームから転送されてきたデータがあります。ホスト上にて作成されたデータであり、かつメインフレームからの転送時にコード変換がされておらず、制御コードが付加されたままであれば、INFILEステートメントでRECFM=オプションに S370V、またはS370VBを指定することで読み込むことが可能です。しかし、一般的なファイル転送ソフトウェアなどで転送された場合、可変長レコードの長さを示すRDWやBDWと呼ばれるフィールドは省略されるため、RECFM=オプションは利用できません。各レコードの長さが格納されたフィールドがない場合、データのレコードフォーマットをRECFM=N(unbufferd)として扱い、次に記載するような方法を利用することで読み込むことが可能です。

- ・レコードの各フィールドを全てフォーマット指定して、フォーマット入力する
- ・レコードの各フィールドを、ポインタコントロールを利用してポインタ入力する
- ・(識別レコードの組み合わせごとに長さが固定であれば)レコードの各フィールドの読み込み位置に、レコードの長さを付与する

可変長レコードの読み込みについては、次のURLで「ホストからのEBCDICデータを読み込む方法」として紹介しています。

「TS-DOCS TS-642 Reading EBCDIC Files on ASCII Systems」
<http://support.sas.com/techsup/technote/ts642.html>

なお、一例として簡単なサンプルを記載します。

前提条件

- ・先頭1バイトのフラグで識別される長さの異なるデータがある
- ・フラグで識別された各オブザベーション(レコード)の組み合わせの長さは固定
- 先頭1バイトが「1」の場合は200バイト
- 先頭1バイトが「2」であれば1000バイト
- ・各オブザベーションには改行コードが存在しない
- ・データはバック十進データを除き、ASCII変換済み

プログラム概要

- ・入力ファイル「VBREC1」をストリームレコード(1レコードのデータ)として読み込み、レコードレイアウトの異なる2つのデータを、それぞれ「REC1」「REC2」データセットとして分けて作成する。
- ・次のデータレコードの読み込み位置を変数「strec」に保持し、該当のレコードの開始位置を特定させる。

```

/* 入力ファイル定義 */
FILENAME vbrec1 'C:\test\testvb.txt';
/* 出力データセット定義 */
DATA WORK.rec1 WORK.rec2;
/* 先頭からの位置を記録した変数をRETAINステートメントで保存 */
RETAIN strec;
/* 1回目のみで初期化 */
IF _n_ = 1 THEN strec = 0 ;
/* RECFM=n指定 */
INFILE vbrec1 RECFM=n ;
/* 1バイトの識別データの読み込み */

```

```

/* 行保持指示子"@"で行を保持 */
INPUT @1+strec w_flg 1. @;
/* 識別変数の選択 */
SELECT(w_flg) ;
/* 条件 =1 */
WHEN('1') DO;
INPUT @2+strec w_num1 4.
@09+strec w_cdata1 $CHAR7.
@16+strec w_cdata2 $CHAR10.
@27+strec w_nopd1 S370FPD4.
@32+strec w_nozd1 3.
@35+strec w_alld1 $CHAR166.
@;
OUTPUT work.rec1;
/* 条件のレコード長指定 */
endrec = 200;
/* STRECへ次のレコード位置格納 */
strec = strec + endrec ;
END;
/* 条件 =2 */
WHEN('2') DO;
INPUT @002+strec w_kdat1 $CHAR818.
@820+strec w_cd1 4. w_cd2 3.
@827+strec w_date YMMDD10.
@838+strec w_nopd1 S370FPD4.
@843+strec w_nozd1 3.
@846+strec w_alld2 $CHAR155.
@;
OUTPUT work.rec2 ;
/* 条件のレコード長指定 */
endrec = 1000;
/* STRECへ次のレコード位置格納 */
strec = strec + endrec ;
END;
OTHERWISE DELETE;
END;
/* DEBUG プログラム作成時のループ防止 */
IF _N_ > 5; STOP;
RUN;

```

Q Hodrick-PrescottフィルターやBaxter-Kingフィルターは、SASでサポートされていますか。

A Hodrick-Prescottフィルターは、SAS9以降ではSAS/ETSのEXPANDプロシジャでサポートされています。TRANSFORMOUT=オプション(またはTRANSFORMIN=オプション)でHP_Tを使用するとトレンド成分が、またHP_Cを使用するとサイクル成分が計算されます。

```

/*SAS9での例*/
PROC EXPAND DATA=test OUT=out FROM=QTR TO=QTR;
  CONVERT y=HP_trend /TRANSFORMOUT=(HP_T 1600); /*HP_T*/
  CONVERT y=HP_cycle /TRANSFORMOUT=(HP_C 1600); /*HP_C*/
RUN;

```

なお、Baxter-King フィルターは、現行リリースではサポートされていません。

Q MULTTESTプロシジャのTESTステートメントでCAを指定し、Cochran-Armitageの傾向性検定を行なっています。このとき、正確な並べ替え分布に基づくp値を算出するためには、PERMUTATION=オプションでどのような値を指定すればよいでしょうか。

A PERMUTATION=オプションで指定している値によって、各層(STRATA)において正確な並べ替え分布を用いるか、あるいは正規分布(continuity-corrected normal distribution)に基づく近似の並べ替え分布を用いるかを判別します。

各層で、応答変数が1である度数と0である度数のうち、最小の値がPERMUTATION=オプションで指定されている値より小さい場合には、正確な並べ替え分布を用います。正確な並べ替え分布に基づくp値を算出するには、全ての層において並べ替え分布が正確である必要があります。この場合、最初に各層における応答変数の水準ごとの度数を求め、小さいほうをその層における「判別値」とします。次に、各層における「判別値」のうち、最も大きい値(またはそれ以上)をPERMUTATION=オプションに指定することになります。

ただし、データセットが非常に大きい場合には、正確な並べ替え分布の作成に多くの時間がかかるため、PERMUTATION=オプションで上限を設定し、正規分布による近似の並べ替え分布も含めた上でのp値の算出を推奨します。



New Publications

新刊マニュアルのお知らせ

「SAS 9.1 Companion for Windows」

注文番号:58942

価格:13,440円(税込)

本書は、Windows版SAS 9.1が提供する機能のうち、Windowsオペレーティングシステムに依存するものをまとめたマニュアルです。本書には、Windows環境におけるSASの起動から各種機能の使用法、Windows版SASに特有のグラフィカルユーザーインターフェイス(GUI)、ステートメント、関数、オプションの詳細が解説されています。それらの機能の利用法として、SASからの電子メール送信や、外部ダイナミックリンク・ライブラリ(DLL)の利用、さらにマイクロソフトパフォーマンス・モニターとイベント・ビューを使ったSASプロセスをモニタリングする方法なども記載されています。Windows版SASを初めてお使いになる方はもちろん、Windows環境に特化したSASの機能を使用するユーザーやシステム管理者の方は、ぜひ本書をお読みください。

「SAS 9.1 Language Reference: Dictionary, Volumes 1, 2, and 3」

注文番号:58941

価格:32,550円(税込)

本書は、SASの中心となるBase SASを構成する各種SASランゲージをアルファベット順に編集した辞書形式のリファレンスで、SASの経験者や、他のプログラミング言語の中級レベル以上のプログラマー向けのマニュアルです。本書では、下記のSASランゲージの概要や使用法について、オペレーショナルシステムに依存しないすべての情報を解説しています。

- ・ SASデータセットオプション
- ・ SAS出力形式 / 入力形式
- ・ SAS関数
- ・ SAS CALLルーチン
- ・ SASステートメント
- ・ Base SASで使用できるステートメントとシステムオプション
- ・ SAS演算子
- ・ DATAステップデバッグ

本書の姉妹編となる「SAS 9.1 Language Reference: Concepts」では、SASの機能、DATAステップ、そしてSASファイルの重要コンセプトをまとめています。

「SAS 9.1 Language Reference: Concepts」

注文番号:58940

価格:13,440円(税込)

本書は、SAS関数、DATAステップ、そしてSASファイルといった、SASにおいて最も重要なコンセプトについて包括的に解説しています。本書には以下のような内容が記載されています。

一般的なBase SASコンセプト

- ・ SASでの処理
- ・ SAS名
- ・ 欠損値
- ・ 時間値と日付値
- ・ WHERE処理
- ・ SAS出力
- ・ エラー処理とデバッグ

Base SASのDATAステップ・コンセプト

- ・ DATAステップ
- ・ BYグループ
- ・ ARRAY処理
- ・ SASデータセットの結合と修正
- ・ DATAステップ・コンポーネント・オブジェクトの使用法
- ・ 生データの読み方

SASウインドウ環境コンセプト

- ・ メインウインドウとウインドウ環境でのデータ管理

Base SASファイル・コンセプト

- ・ SASデータライブラリ
- ・ SASデータセット
- ・ SASカタログ
- ・ スタアドプログラム
- ・ インデックス
- ・ 整合性制約
- ・ 監査証跡
- ・ バージョン間の互換性
- ・ ファイル保護
- ・ SAS I/Oエンジン
- ・ SASファイル管理
- ・ 外部ファイル

SASで使用されるプロトコル

- ・ SMTP電子メールとUUID

このリファレンスマニュアルは、「SAS 9.1 Language Reference: Dictionary」の姉妹編です。

「SAS/STAT 9.1 User's Guide, Volumes 1-7」

注文番号:59051

価格:74,655円(税込)

本書は、SAS/STATで使用できるプロシジャの最新リファレンスマニュアルで、下記の各種統計手法をカバーしています。

- ・ 分散分析
- ・ 回帰
- ・ カテゴリカルデータ解析
- ・ 多変量解析
- ・ 生存時間解析
- ・ クラスタ分析
- ・ ノンパラメトリック解析
- ・ 調査データ分析

それぞれのプロシジャについて、構文、使用法に関する情報、サンプルプログラムが記載され、ODS利用についての解説も含まれています。ほぼ全編にわたりSAS 9.1での拡張に合わせて改訂が施されています。SAS 9.1の新機能紹介の章では、検定力やサンプルサイズの計算、ロバスト回帰、多重代入法、そして調査データのロジスティック回帰などについて解説しています。

「SAS 9.1.3 High-Performance Forecasting: User's Guide」

注文番号:59877

価格:9,660円(税込)

本書は、High-Performance Forecastingの構文とその使用方法について、用例に基づき解説したものです。

「SAS 9.1.3 XML LIBNAME Engine: User's Guide」

注文番号:59888

価格:3,885円(税込)

本書では、XMLエンジンを使用してSASデータセットをXML文書にエクスポートする方法、また外部XML文書をSASデータセットとしてインポートする方法について解説しています。SASやXMLの知識があり、XML文書を扱うユーザー向けのマニュアルです。

上記以外にもSAS9に関するマニュアルを多数ご用意しております。詳しくはhttp://www.sas.com/japan/manual/20040810_v9.pdfをご参照ください。

マニュアル販売係

- TEL 03-3533-3835
- FAX 03-3533-3781
- E-mail JPNBooksale@sas.com

SASマニュアル注文用紙、および最新のPublication Catalog (マニュアル案内パンフレット)は弊社ホームページ<http://www.sas.com/japan/manual/>にて公開していますので、併せてご利用ください。

SAS Training

SASトレーニングのお知らせ

新規コースのご案内

「医薬向けSASによる生存時間解析」コース

日 程: 2004年12月17日(金)10:00 ~ 17:00(東京会場)

価 格: 47,250円(税込)/チケット捺印数1

受講対象: SASによる統計解析に従事している方

前提知識: 「医薬向けSASによるデータ解析入門」を受講済みか、同程度の知識のある方。

具体的には、以下の知識を前提としています。

- ・簡単なDATAステッププログラムが作成できる
- ・分割表解析、Cochran-Mantel-Haenszel法、カイニ乗検定の知識
- ・基本的な回帰分析(できれば重回帰分析)の知識

学習内容: 生存時間解析の基礎を解説し、実際の医学データの解析を通してLIFETESTプロシジャ・PHREGプロシジャの使い方を紹介します。

- ・生存時間解析の基本概念
- ・生存関数のノンパラメトリックな推定 - Kaplan-Meier法
- ・生存関数のノンパラメトリックな検定 - Log-rank検定、一般化Wilcoxon検定
- ・Cox比例ハザードモデル

「SAS Enterprise Guide操作入門」コース

日 程: 日程未定、13:00 ~ 17:00(東京会場)

通常のコースとは開催時間が異なりますのでご注意ください。

価 格: 31,500円(税込)/チケット捺印数1

受講対象: SAS Enterprise Guideを使用して、ポイント&クリックでSASの機能を活用したい方

前提知識: マウス操作、および標準的なコンピュータの利用方法

学習内容: SAS Enterprise Guideの基本的な操作方法の説明を中心に、データ入力、データ加工、データ集計、レポート作成、グラフ作成など、マウス操作で実現するSAS Enterprise Guideのさまざまな機能を習得します。

「SAS 9.1 概要」コース

日 程: 日程未定、14:00 ~ 17:00(東京会場)

通常のコースとは開催時間が異なりますのでご注意ください。

価 格: 31,500円(税込)/チケット捺印数1

受講対象: SASの最新リリース9.1が実現する機能や拡張点に興味のある方、また導入をお考えの方

前提知識: 特にありません

学習内容: SAS 9.1のインストール時の注意点から、従来製品の簡単な拡張点紹介、および新製品としての各種サーバー製品の紹介を行います。

- ・SAS 9.1の概要説明
拡張点の概要、SASエクスプローラの拡張点、SAS管理コンソール
- ・SAS 9.1の環境
SAS 9.1 Foundationのインストールと環境、National Language Support
- ・Base SAS の各プロシジャの拡張点
マルチスレッド、MIGRATEプロシジャ

・SAS言語の拡張点

SASシステムオプション、SAS関数、SASフォーマットとインフォーマット、演習問題

・SAS/ACCESSの拡張点

LIBNAMEエンジン、データのインポート、データのエクスポート

「SAS 9.1 拡張点」コース

日 程: 日程未定、10:00 ~ 17:00(東京会場)

価 格: 47,250円(税込)/チケット捺印数1

受講対象: SASプログラムを記述してSASを利用している方

前提知識: マウス操作、および標準的なコンピュータの利用方法

SAS9以前のリリースのSASに関する知識のある方

学習内容: 下記の内容についてPCを使用して実習します。

- ・SAS 9.1の概要説明
- ・SAS 8.2からの移行について
- ・SAS プロシジャの拡張点
マルチスレッドおよび並列処理
- ・SAS言語の拡張点
SASシステムオプション、SAS関数、Perl正規表現、ODS拡張点、XMLによるデータ処理、演習問題
- ・SPDエンジンの使用方法
- ・SAS/ACCESSの拡張点
LIBNAMEエンジン、データのインポート、データのエクスポート
- ・MP Connectの拡張点
- ・補足資料として、次の製品の紹介
SAS Enterprise Guide、SAS/STAT拡張点、マクロの拡張点

プロフェッショナルコース開催のご案内

「決定木モデル」コース

日 程: 11月17日(水)~ 18日(木)10:30 ~ 16:30(東京会場)

価 格: 94,500円(税込)/チケット捺印数2

受講対象: データマイニングを業務で活用される方

前提知識: 「SASプログラミング Basic I」コースを受講済みか、同程度の知識のある方

- ・一般的な統計の知識のある方

学習内容: データマイニング技法のひとつである、決定木(Decision Tree)について比較的経験の浅いユーザーを対象に、SAS Enterprise Miner 4.1を用いて解説します。必要とされる基礎概念についても、ひとつおりの復習・解説しますので、SASの利用と統計に関する初歩的知識を除いて、同技法に関する予備知識は必要ありません。さまざまな事例を検討したり、SASを用いた演習を交えることによって、初めて同技法を学ぶ方が、各手法を実践的に利用できる能力を身につけることに焦点をあてています。

- ・決定木の目的と概要
- ・基礎的概念の復習: 相関比、平方和、F検定、X2検定、ジニ係数、情報量
- ・決定木の諸手法(CHAID、C&RT、C4.5)
- ・モデルの精緻化、検証、精度分析
- ・モデルの利用

今後も多岐にわたったトレーニングコースを追加していく予定です。日程等の詳細は弊社Webサイトに順次公開しますので、以下のURLをご参照ください。

<http://www.sas.com/japan/training/>

トレーニング担当

- T E L 03-3533-3835
- F A X 03-3533-3781
- E-mail JPNTraining@sas.com

Programmers Workshop

SASプログラマーズ・ ワークショップ開催のお知らせ

今回のワークショップでは、SASの最新リリースであるSAS 9.1.3をご紹介します。今号のTechnical Newsで紹介しきれていない新機能や、実際に移行する上での注意点などについて、デモを交えながら解説します。

	東京会場	大阪会場
日時	2004年11月19日(金) 15:00 ~ 17:00	2004年11月26日(金) 15:00 ~ 17:00
会場	SAS Institute Japan 株式会社 東京本社 プレゼンテーションルーム 東京都中央区勝どき1-13-1 イヌイビル・カチドキ8F	SAS Institute Japan 株式会社 大阪支店 プレゼンテーションルーム 大阪市北区堂島浜1-4-16 アクア堂島西館12F
定員	各回とも50名	
参加費	無料	

両日とも参加は無料です。定員に限りがございますので、お早めにお申し込みください。お申し込みはFAXまたはE-mailで承ります。「SASプログラマーズ・ワークショップ参加申込用紙」に必要事項をご記入の上、下記連絡先にご送付ください。お申し込み用紙は下記のURLより入手できます。

<http://www.sas.com/japan/service/technical/pws.html>

その他、ご不明点、ご質問がございましたら、下記までご連絡ください。

SASプログラマーズ・ワークショップ担当

T E L 03-3533-3877

F A X 03-3533-3781

E-mail JPNsaspws@sas.com

Latest Releases

最新リリース情報

PCプラットフォーム

Windows版	SAS 9.1.3	9.1 TS1M3
Windows(64-bit)版	SAS 9.1.3	9.1 TS1M3

ミニコンピュータプラットフォーム

OpenVMS AXP版	SAS 6.12	TS020
OpenVMS VAX版	SAS 6.08	TS407

UNIXプラットフォーム

Tru64版	SAS 9.1.3	9.1 TS1M3
SunOS/Solaris版	SAS 9.1.3	9.1 TS1M3
HP-UX版	SAS 9.1.3	9.1 TS1M3
HP-UX(Itanium)版	SAS 9.1.3	9.1 TS1M3
AIX版	SAS 9.1.3	9.1 TS1M3
Linux(Intel)版	SAS 9.1.3	9.1 TS1M3
ABI+版	SAS 6.11	TS040

メインフレームプラットフォーム

IBM版(OS/390, z/OS)	SAS 8.2	TS2M0
富士通版(F4, MSP)	SAS 6.09E	TS470
日立版(VOS3)	SAS 6.09E	TS470
CMS版	SAS 6.08	TS410

SAS Technical News Autumn 2004

発行
SAS Institute Japan株式会社

テクニカルニュースに関するお問い合わせ先

テクニカルサポートグループ

TEL: 03-3533-3877

FAX: 03-3533-3781

E-mail: JPNTechnews@sas.com



SAS Institute Japan株式会社 www.sas.com/japan/

このカタログに記載された内容は改良のため、予告なく仕様・性能を変更する場合があります。あらかじめご了承ください。
SASロゴ、The Power to Knowは米国SAS Institute Inc.の登録商標です。その他記載のブランド、商品名は、一般に各社の登録商標です。Copyright©2004, SAS Institute Inc. All rights reserved.

東京本社
〒104-0054
東京都中央区勝どき1-13-1
イヌイビル・カチドキ
Tel 03 (3533) 6921
Fax 03 (3533) 6927

大阪支店
〒530-0004
大阪市北区堂島浜1-4-16
アクア堂島西館 12F
Tel 06 (6345) 5700
Fax 06 (6345) 5655